

## Solution to Exercise 4: Adding field labels and value labels

### Key Point(s):

- A Field label helps the data entry person to become familiar with the definition of the fields.
- A Value label helps the data entry person to see Field values and Value labels. This is particularly helpful when the Field values are numeric.

### Task

- o *Revise the A\_EX03.QES file, save it as A\_EX04.QES, make the REC file, open the A\_EX03.CHK file, save it as A\_EX04.CHK and make the necessary changes and then enter some made-up data to check the functionality.*

### Solution

The data entry form looks as follows (**CTRL+T**):

This is the questionnaire for the laboratory register

serno	Laboratory serial number	<input type="text"/>	Write note (F5) if alternative number must be used
regdate	Registration date	<input type="text"/>	Enter 01/01/1800 if not recorded
sex	Examinee's sex	<input type="text"/>	
age	Examinee's age in years	<input type="text"/>	Enter 999 if not recorded
reason	Examination reason	<input type="text"/>	
res1	Result of specimen 1	<input type="text"/>	
res2	Result of specimen 2	<input type="text"/>	
res3	Result of specimen 3	<input type="text"/>	

You note that in this solution we have added more to the QES file than was asked for in the **Task**: information was written into the QES file to the left of the field, but only for some fields.

### When to write instructions into the QES file?

The fields `SEX`, `REASON`, `RES1`, `RES2`, and `RES3` all have Labelblocks. Thus, all the instructions what can be entered is provided. Furthermore, the Value label is written into the space right to the field. There is thus no need to write an instruction nor would it be sensible to do so. On the other hand, the fields `SERNO`, `REGDATE`, and `AGE` do not have Labelblocks and the data entry person will thus not know (before making an error when the message says what is legal) what to do if there is a duplicate `SERNO` (a non-unique identifier) or if the registration date is missing. It is thus a good idea to provide for fields which do not have Labelblocks instructions directly in the QES file.

And this is the `A_EX04.CHK` file (**CTRL+O**, pick EpiData check files as Type of files):

LABELBLOCK

```

LABEL label_sex
  1 Female
  2 Male
  9 "No sex recorded"
END
LABEL label_reason
  0 Diagnosis
  8 "Follow-up, month not stated"
  9 "Reason not stated"
  1 "Follow-up at 1 month"
  2 "Follow-up at 2 months"
  3 "Follow-up at 3 months"
  4 "Follow-up at 4 months"
  5 "Follow-up at 5 months"
  6 "Follow-up at 6 months"
  7 "Follow-up at 7 months or later"
END
LABEL label_result
  0.0 Negative
  1.0 "1+ positive"
  2.0 "2+ positive"
  3.0 "3+ positive"
  4.0 "4+ positive"
  9.0 "No result recorded"
  5.0 "Positive, not quantified"
  6.0 "Scanty, not quantified"
  0.1 "Scanty, 1 AFB per 100 fields"
  0.2 "Scanty, 2 AFB per 100 fields"
  0.3 "Scanty, 3 AFB per 100 fields"
  0.4 "Scanty, 4 AFB per 100 fields"
  0.5 "Scanty, 5 AFB per 100 fields"
  0.6 "Scanty, 6 AFB per 100 fields"
  0.7 "Scanty, 7 AFB per 100 fields"
  0.8 "Scanty, 8 AFB per 100 fields"
  0.9 "Scanty, 9 AFB per 100 fields"
END
END

serno
  MUSTENTER
  KEY UNIQUE
END

regdate
  RANGE 01/01/2000 31/12/2005
  LEGAL
  01/01/1800
  END
  MUSTENTER
END

sex
  COMMENT LEGAL USE label_sex SHOW
  MUSTENTER
  TYPE COMMENT
END

age
  RANGE 0 125
  LEGAL
  999
  END
  MUSTENTER
END

reason
  COMMENT LEGAL USE label_reason SHOW
  MUSTENTER

```

```
TYPE COMMENT
END

res1
COMMENT LEGAL USE label_result SHOW
MUSTENTER
TYPE COMMENT
END

res2
COMMENT LEGAL USE label_result SHOW
MUSTENTER
TYPE COMMENT
END

res3
COMMENT LEGAL USE label_result SHOW
MUSTENTER
TYPE COMMENT
END

AFTER RECORD
IF serno=. THEN
HELP "Core information missing:\n SERNO\n must be available" TYPE=WARNING
GOTO serno
ENDIF
END
```